Computations for the Hopf Bifurcation in Wright’s Equation

This document presents the content from the following sections:

Section 2: Preliminaries

Linear Operator Definitions

We define the operator K as KK, and it is defined to operate on just the first 5 terms of a sequence. (The variable K cannot be used as it is a globally defined constant.)

In[1]:=
```
KK = DiagonalMatrix[Table[1/k, {k, 1, 5}]];
```

We define the operator U_ω as $U[\omega]$, and it is defined to operate on just the first 5 terms of a sequence.

In[2]:=
```
U[w_] := DiagonalMatrix[Table[Exp[-I k w], {k, 1, 5}]]
```

We define the shift operators σ^- and σ^+ as Sn and Sp respectively.
These are only defined to take the first 4 terms of a sequence as input.

In[3]:=
```
Sn = DiagonalMatrix[Table[1, {k, 1, 4}], 1];
Sp = DiagonalMatrix[Table[1, {k, 1, 4}], -1];
```

We define the operator L_ω as $L[\omega]$. This is defined to operate on just the first 5 terms of a sequence, for which the first sequence term is exactly zero.
WARNING: Do not multiply L with something with non-zero F_1 term.

In[5]:=
```
L[w_] := (Sp.(Exp[-I w] IdentityMatrix[5] + U[w]) + Sn.(Exp[I w] IdentityMatrix[5] + U[w]))
```

We define L_{ω_0} as `L0`.

In[6]:=
```
L0 := L[π / 2]
```

Definition 2.10 --- Approximate Solution

We define the approximate solution $(\bar{\alpha}_\epsilon, \bar{\omega}_\epsilon, \bar{c}_\epsilon)$ as `(a,w,c2)`.

In[7]:=
$$a[\epsilon\_] := \pi / 2 + \frac{\epsilon^2}{5} (3 \pi / 2 - 1)$$
$$w[\epsilon\_] := \pi / 2 - \frac{\epsilon^2}{5}$$
$$c2[\epsilon\_] := \frac{(2 - I)}{5} \epsilon$$

Section 3: Local Results

Section 3.1: Constructing a Newton-Like Operator

Definition 3.1

In Definition 3.1 we define

$$A \; := A_0 + \epsilon A_1$$
$$A^\dagger := A_0^{-1} - \epsilon A_0^{-1} A_1 A_0^{-1}$$

and

$$A_0(\alpha, \omega, c) := i_{\mathbb{C}} A_{0,1} \begin{bmatrix} \alpha \\ \omega \end{bmatrix} e_1 + A_{0,*} c$$

$$A_1(\alpha, \omega, c) := i_{\mathbb{C}} A_{1,2} \begin{bmatrix} \alpha \\ \omega \end{bmatrix} e_2 + A_{1,*} c$$

In this Mathematica notebook, we do not explicitly define the operators A and $A^\dagger$. We do however define the component operators of A_0, A_0^{-1} and A_1.

We define the map $A_{0,1}$ as **A01**.

In[10]:=
$$A01 = \begin{pmatrix} 0 & -\pi / 2 \\ -1 & 1 \end{pmatrix};$$

We define the map $A_{1,2}$ as **A12**.

In[11]:=
$$A12 = \frac{1}{5} \begin{pmatrix} -2 & (4 - 3 \pi) / 2 \\ -4 & 2 (2 + \pi) \end{pmatrix};$$

We define the map $A_{0,*}$ as **A0s.**

In[12]:=
$$A0s = \frac{\pi}{2} (I \, Inverse[KK] + U[\pi / 2]);$$

We define the map $A_{1,*}$ as **A1s**.

In[13]:=
$$A1s = \frac{\pi}{2} L0;$$

Proposition 3.2

In[14]:= $\left(\frac{2\sqrt{10}}{5}\right)^{-1}$ // N

Out[14]= 0.790569

Appendix A: Operator Norms

We define the operator $\hat{U}$ as **Uhat**, and it is defined to operate on just the 2^{nd} – 5^{th} terms of a sequence.

In[15]:= **Uhat = DiagonalMatrix**$\left[\text{Table}\left[\text{If}\left[k == 1, 0, \frac{1}{1 - I\, k^{-1}\, \text{Exp}[-I\, k\, \pi / 2]}\right], \{k, 1, 5\}\right]\right]$;

We define $A_{0,*}{}^{-1} := \frac{2}{i\pi}\hat{U}$ K as **A0sI**

In[16]:= **A0sI =** $\frac{2}{I\,\pi}$ **Uhat.KK;**

For future use, we define $(A_{0,1})^{-1}$ as **A01I.**

In[17]:= **A01I = Inverse[A01]**

Out[17]= $\left\{\left\{-\frac{2}{\pi}, -1\right\}, \left\{-\frac{2}{\pi}, 0\right\}\right\}$

Proposition A.1

We check that the norm or $||\hat{U}$ c$||$ is maximized when $c = e_5$

In[18]:= **Table**$\left[\text{Norm}\left[\frac{1}{1 - I\, k^{-1}\, \text{Exp}[-I\, k\, \pi / 2]}\right], \{k, 2, 10\}\right]$

Out[18]= $\left\{\frac{2}{\sqrt{5}}, \frac{3}{4}, \frac{4}{\sqrt{17}}, \frac{5}{4}, \frac{6}{\sqrt{37}}, \frac{7}{8}, \frac{8}{\sqrt{65}}, \frac{9}{8}, \frac{10}{\sqrt{101}}\right\}$

We check that the norm or $||\hat{U}$ K c$||$ is maximized when $c = e_2$.

In[19]:= **Table**$\left[\frac{1}{k}\,\text{Norm}\left[\frac{1}{1 - I\, k^{-1}\, \text{Exp}[-I\, k\, \pi / 2]}\right], \{k, 2, 10\}\right]$

Out[19]= $\left\{\frac{1}{\sqrt{5}}, \frac{1}{4}, \frac{1}{\sqrt{17}}, \frac{1}{4}, \frac{1}{\sqrt{37}}, \frac{1}{8}, \frac{1}{\sqrt{65}}, \frac{1}{8}, \frac{1}{\sqrt{101}}\right\}$

Proposition A.2

We verify an inequality.

In[20]:= $\text{Max}\left[\left\{\frac{1}{5}\sqrt{\frac{45+5\sqrt{17}}{2}}, \frac{2\sqrt{10}}{5}\right\}\right]$

Out[20]= $2\sqrt{\frac{2}{5}}$

Proposition A.3 -- An upper bound on: $A_0^{-1} A_1$

We define an upper bound on $A_0^{-1} A_1$ as **A0A1.**

In[21]:= $\text{A0A1} = \begin{pmatrix} 0 & 0 & \frac{1}{2}\sqrt{2+\pi^2/2} \\ 0 & 0 & \frac{1}{\sqrt{2}} \\ \frac{8}{5\pi} & \frac{2\sqrt{16+8\pi+5\pi^2}}{5\pi} & \frac{2}{\sqrt{5}} \end{pmatrix};$

Appendix B: Endomorphism on a Compact Domain

Various Bounds

Definition B.1

We define Δ_α, Δ_ω and δ_c as **Da, Dw and δ.**

In[22]:= $\text{Da}[\epsilon\_, \text{ra}\_] := \frac{\epsilon^2}{5}\left(3\frac{\pi}{2}-1\right)+\text{ra}$

$\text{Dw}[\epsilon\_, \text{rw}\_] := \frac{\epsilon^2}{5}+\text{rw}$

$\delta[\epsilon\_, \text{rc}\_] := \frac{2}{\sqrt{5}}\epsilon+\text{rc}$

The variables Δ_α^0, Δ_ω^0 and δ_c^0 are given as **Da[ϵ,0], Dw[ϵ,0] and δ0[ϵ].**

In[25]:= $\delta 0[\epsilon\_] := \delta[\epsilon, 0]$

Proposition B.4 - Rho

Used in the definition of C_0, we define $\overline{\hat{U} A_1}$ as **UA1.**

In[26]:= $\text{UA1} = \left\{\frac{8}{5}, \frac{2}{5}\sqrt{16+8\pi+5\pi^2}, \frac{5\pi}{2}\right\};$

We define C_0, C_1, C_2, and C_3 as **C0, C1, C2 and C3.**

In[27]:=
$$\text{C0}[\epsilon\_, \text{da}\_, \text{dw}\_, \text{rc}\_] := \frac{2\,\epsilon^2}{\pi}\,\text{UA1.A0A1.}\{0, 0, \delta[\epsilon, \text{rc}]\}$$

$$\text{C1}[\epsilon\_] := \frac{5}{2\,\pi} + \frac{\epsilon\,\sqrt{10}}{\pi}$$

$$\text{C2}[\epsilon\_, \text{rw}\_] := \text{Dw}[\epsilon, \text{rw}]\,\left(\left(1 + \frac{\pi}{2}\right) + \epsilon\,\pi\right)$$

$$\text{C3}[\epsilon\_, \delta\_, \text{da}\_, \text{dw}\_] := \text{da}\,(2 + \delta) + 2\,\text{dw}\,\left(1 + \frac{\pi}{2}\right) + \epsilon\,\left(\pi + 2\,\text{da} + 4\,\delta\,\text{da} + \pi\,\text{dw}\,\delta + \left(\frac{\pi}{2} + \text{da}\right)\delta^2\right)$$

We define the function $C(\epsilon, r_\alpha, r_\omega, r_c)$ as **ρ.**

In[31]:=
$$\rho[\epsilon\_, \text{ra}\_, \text{rw}\_, \text{rc}\_] := \frac{1}{1 - \text{C1}[\epsilon]\,\text{C2}[\epsilon, \text{rw}]}$$
$$(\text{C0}[\epsilon, \text{Da}[\epsilon, \text{ra}], \text{Dw}[\epsilon, \text{rw}], \text{rc}] + \text{C1}[\epsilon]\,\text{C3}[\epsilon, \delta[\epsilon, \text{rc}], \text{Da}[\epsilon, \text{ra}], \text{Dw}[\epsilon, \text{rw}]])$$

Appendix C: The bounding functions for Y(ϵ)

Y Bounds

These are the functions $F(x_\epsilon)$ exactly.

These functions are defined, approximated, but not used.

In[32]:=
```
F₁[ϵ_, a_, w_, c2_] := (I w + a Exp[-I w]) + a ϵ (Exp[I w] + Exp[-2 I w]) c2
F₂[ϵ_, a_, w_, c2_] := (2 I w + a Exp[-2 I w]) c2 + a ϵ Exp[-I w]
F₃[ϵ_, a_, w_, c2_] := a ϵ (Exp[-I w] + Exp[- 2 I w]) c2
F₄[ϵ_, a_, w_, c2_] := a ϵ Exp[-2 I w] c2^2
```

In[36]:=
```
F1[ϵ_] := F₁[ϵ, a[ϵ], w[ϵ], c2[ϵ]]
F2[ϵ_] := F₂[ϵ, a[ϵ], w[ϵ], c2[ϵ]]
F3[ϵ_] := F₃[ϵ, a[ϵ], w[ϵ], c2[ϵ]]
F4[ϵ_] := F₄[ϵ, a[ϵ], w[ϵ], c2[ϵ]]
```

In[40]:=
```
F[ϵ_] := Table[If[i ≥ 5, 0, Fᵢ[ϵ, a[ϵ], w[ϵ], c2[ϵ]]], {i, 1, 5}]
```

The polynomial Approximations of $F(x_\epsilon)$

Proposition C.2 is delayed until the end of this section.

Proposition C.3 --- Bound on f_1

In[41]:=
$$f_1[\epsilon\_] := \frac{\pi}{2}\left(\frac{1}{2}\,\mathrm{Dw}[\epsilon, 0]^2 + \frac{1}{6}\,\mathrm{Dw}[\epsilon, 0]^3\right) + \mathrm{Da}[\epsilon, 0]\,\mathrm{Dw}[\epsilon, 0] + \mathrm{Da}[\epsilon, 0]\,\epsilon\,\delta[\epsilon, 0] + \frac{3\pi}{4}\,\mathrm{Dw}[\epsilon, 0]\,\epsilon\,\delta[\epsilon, 0]$$

Proposition C.4 --- Bound on f_2

In[42]:=
$$f_2[\epsilon\_] := \left(\frac{\pi}{2} + \mathrm{Da}[\epsilon, 0]\right)\mathrm{Dw}[\epsilon, 0]\,(\delta[\epsilon, 0] + \epsilon) + \frac{1}{2}\,\delta[\epsilon, 0]\,(2\,\mathrm{Dw}[\epsilon, 0] + \mathrm{Da}[\epsilon, 0]) + \epsilon\,\mathrm{Da}[\epsilon, 0]$$

Proposition C.5 --- Bound on f_3

In[43]:=
$$f_3[\epsilon\_] := \frac{1}{2}\left(\frac{\pi}{2} + \mathrm{Da}[\epsilon, 0]\right)\left(\sqrt{2} + 3\,\mathrm{Dw}[\epsilon, 0]\right)\epsilon\,\delta[\epsilon, 0]$$

Proposition C.6 --- Bound on f_4

In[44]:=
$$f_4[\epsilon\_] := \frac{1}{5}\left(\frac{\pi}{2} + \mathrm{Da}[\epsilon, 0]\right)\epsilon^3$$

We define all the functions f_i as a single function

In[45]:=
```
f[ϵ_] := Table[If[i ≥ 5, 0, f_i[ϵ]], {i, 1, 5}]
```

Theorem C.2 -- We compute a polynomial upper bound on $A^\dagger F(x_\epsilon)$

We recall that

$$A^\dagger := A_0^{-1} - \epsilon A_0^{-1} A_1 A_0^{-1}$$

and

$$A_0(\alpha, \omega, c) := i_{\mathbb{C}} A_{0,1}\begin{bmatrix}\alpha\\ \omega\end{bmatrix} e_1 + A_{0,*}\,c$$

$$A_1(\alpha, \omega, c) := i_{\mathbb{C}} A_{1,2}\begin{bmatrix}\alpha\\ \omega\end{bmatrix} e_2 + A_{1,*}\,c$$

In order to compute $A^\dagger \mathrm{F}(\epsilon)$, we break up the computation into its first and second order terms.

First Order Terms

We calculate the bound on $A_0^{-1} F(\epsilon)$.

In[46]:=
```
FirstOrderAW[ϵ_] := (Abs[A01I].{1, 1}) f_1[ϵ]
FirstOrderC[ϵ_] := Abs[A0sI].f[ϵ]
```

Second Order Terms

We may write

$$\begin{aligned} A_0{}^{-1} A_1 A_0{}^{-1} &= \left(A_{0,1}{}^{-1} + A_{0,*}{}^{-1}\right)\left(A_{1,2} + A_{1,*}\right)\left(A_{0,1}{}^{-1} + A_{0,*}{}^{-1}\right) \\ &= A_{0,1}{}^{-1} A_{1,*} A_{0,*}{}^{-1} + A_{0,*}{}^{-1} A_{1,2} A_{0,1}{}^{-1} + A_{0,*}{}^{-1} A_{1,*} A_{0,*}{}^{-1} \end{aligned}$$

We compute the $A_{0,1}{}^{-1} A_{1,*} A_{0,*}{}^{-1}$ term.

In[48]:=
```
SecondOrderAW[ϵ_] := Abs[A01I].(Abs[A1s.A0sI][[1, 2]] {1, 1}) f2[ϵ]
```

We compute the $A_{0,*}{}^{-1} A_{1,2} A_{0,1}{}^{-1}$ term.

In[49]:=
```
SecondOrderC2[ϵ_] := Abs[A0sI[[2, 2]] ] Abs[{1, I}.A12.A01I.{ 1, 1}] f1[ϵ]
```

We compute the $A_{0,*}{}^{-1} A_{1,*} A_{0,*}{}^{-1}$ term.

In[50]:=
```
SecondOrderC[ϵ_] := Abs[A0sI.A1s.A0sI].f[ϵ]
```

Total $A^{\dagger} F(\epsilon)$ = Y Bound

In[51]:=
```
AW[ϵ_] := FirstOrderAW[ϵ] + ϵ SecondOrderAW[ϵ]
Cs[ϵ_] := FirstOrderC[ϵ] + ϵ ({0, SecondOrderC2[ϵ], 0, 0, 0} + SecondOrderC[ϵ])
```

We define the Y functions

In[53]:=
```
Ya[ϵ_] := AW[ϵ][[1]]
Yw[ϵ_] := AW[ϵ][[2]]
Yc[ϵ_] := 2 Total[Cs[ϵ]]
```

Appendix D: The bounding functions for Z(ϵ,r,ρ)

Z Bounds

We delay Theorem D.1 until the end of the section

Proposition D.2

We define the function $f_{1,\alpha}$.

In[56]:=
```
f1a[ϵ_, dw_, δ_] := dw + ϵ δ (2 + δ)/2
```

Proposition D.3

We define the function $f_{1,\omega}$.

In[57]:= $$\text{f1w}[\epsilon\_,\ \text{da}\_,\ \text{dw}\_,\ \delta\_,\ \rho\_] := \text{da} + \frac{\pi}{2}\,\text{dw} + \left(\frac{\pi}{2} + \text{da}\right)\frac{\epsilon\,\delta}{2}\,(3 + \rho)$$

Proposition D.4

We define the function $f_{1,c}$.

In[58]:= $$\text{f1c}[\epsilon\_,\ \text{da}\_,\ \text{dw}\_,\ \delta\_] := \epsilon\left(\text{da} + \frac{3\,\pi}{4}\,\text{dw} + \left(\frac{\pi}{2} + \text{da}\right)\delta\right)$$

Proposition D.5

We define the function $f_{*,\alpha}$.

In[59]:= $$\text{fsa}[\epsilon\_,\ \text{rc}\_,\ \text{dw}\_,\ \delta\_] := \frac{2}{\pi\sqrt{5}}\left(\text{rc} + 2\,\text{dw}\left(\frac{2\,\epsilon}{\sqrt{5}} + \epsilon\right) + \epsilon\,\delta\,(4 + \delta)\right)$$

Proposition D.6

We define the function $f_{*,\omega}$.

In[60]:= $$\text{fsw}[\epsilon\_,\ \text{rc}\_,\ \text{da}\_,\ \text{dw}\_,\ \delta\_,\ \rho\_] := \frac{5}{2\,\pi}\left(1 + \frac{\pi}{2}\right)\text{rc} + \frac{2}{\sqrt{5}}\,\epsilon\left(\left(1 + \frac{4}{\sqrt{5}}\right)\text{dw} + \frac{2}{\pi}\,\text{da}\right) + \frac{5}{2\,\pi}\,\text{da}\,(\text{rc} + \delta) + \frac{2}{\pi}\,\epsilon\left(\frac{\pi}{2} + \text{da}\right)\left(\frac{1}{\sqrt{5}}\,(\delta + \text{rc}) + \frac{5}{4}\left(\delta + \frac{3}{2}\,\text{rc}\right) + \rho\,\frac{\delta}{\sqrt{5}}\right)$$

Proposition D.7

We define the function $f_{*,c}$.

In[61]:= $$\text{fsc}[\epsilon\_,\ \text{da}\_,\ \text{dw}\_,\ \delta\_] := \left(\frac{5}{2}\left(\frac{1}{2} + \frac{1}{\pi}\right)\text{dw} + \frac{\text{da}}{\sqrt{5}}\right) + \epsilon\left(\frac{8}{\pi\sqrt{5}}\,\text{da} + \left(\frac{2}{\sqrt{5}} + \frac{25}{8}\right)\text{dw} + \frac{4\left(\frac{\pi}{2} + \text{da}\right)\delta}{\pi\sqrt{5}}\right)$$

Theorem D.1

We define the matrix-valued function M as below.

In[62]:= $$\text{M}[\epsilon\_,\ \text{ra}\_,\ \text{rw}\_,\ \text{rc}\_,\ \rho\_] := \begin{pmatrix} \sqrt{1 + \frac{4}{\pi^2}} & 0 \\ \frac{2}{\pi} & 0 \\ 0 & 1 \end{pmatrix} .$$

$$\begin{pmatrix} \text{f1a}[\epsilon,\ \text{Dw}[\epsilon,\ \text{rw}],\ \delta[\epsilon,\ \text{rc}]] & \text{f1w}[\epsilon,\ \text{Da}[\epsilon,\ \text{ra}],\ \text{Dw}[\epsilon,\ \text{rw}],\ \delta[\epsilon,\ \text{rc}],\ \rho] & \text{f1c}[\epsilon,\ \text{Da}[\epsilon, \\ \text{fsa}[\epsilon,\ \text{rc},\ \text{Dw}[\epsilon,\ \text{rw}],\ \delta[\epsilon,\ \text{rc}]] & \text{fsw}[\epsilon,\ \text{rc},\ \text{Da}[\epsilon,\ \text{ra}],\ \text{Dw}[\epsilon,\ \text{rw}],\ \delta[\epsilon,\ \text{rc}],\ \rho] & \text{fsc}[\epsilon,\ \text{Da}[\epsilon, \end{pmatrix}$$

We define the upper bound Z as below.

In[63]:=
```
Z[ϵ_, ra_, rw_, rc_, ρ_] := ϵ^2 A0A1.A0A1 + (IdentityMatrix[3] + ϵ A0A1 ).M[ϵ, ra, rw, rc, ρ]
```

Section 3: Local Results

Section 3.2: Explicit Contraction Bounds

Theorem 3.7 and the definition of radii polynomials

In[64]:=
```
Pa[ϵ_, ra_, rw_, rc_, ρ_] :=
  Ya[ϵ] + {1, 0, 0}.(Z[ϵ, ra, rw, rc, ρ] - IdentityMatrix[3]).{ra, rw, rc}
Pw[ϵ_, ra_, rw_, rc_, ρ_] :=
  Yw[ϵ] + {0, 1, 0}.(Z[ϵ, ra, rw, rc, ρ] - IdentityMatrix[3]).{ra, rw, rc}
Pc[ϵ_, ra_, rw_, rc_, ρ_] :=
  Yc[ϵ] + {0, 0, 1}.(Z[ϵ, ra, rw, rc, ρ] - IdentityMatrix[3]).{ra, rw, rc}
```

Proposition 3.10 (a)

We fix a collection of ϵ, r, and ρ for Proposition 3.10 (a).

In[67]:=
```
WrightConjecture =
   {ϵ → Interval[{0.029, 0.029}],
    ra → Interval[{0.13, 0.13}],
    rw → Interval[{0.17, 0.17}],
    rc → Interval[{0.17, 0.17}],
    ρW → Interval[{1.78, 1.78}]};
```

We check that the hypothesis for Proposition B.4 is satisfied

In[68]:=
```
ρW > ρ[ϵ, ra, rw, rc] /. WrightConjecture
```

Out[68]=
```
True
```

We calculate all of the radii polynomials.

In[69]:=
```
Pa[ϵ, ra, rw, rc, ρW] /. WrightConjecture
Pw[ϵ, ra, rw, rc, ρW] /. WrightConjecture
Pc[ϵ, ra, rw, rc, ρW] /. WrightConjecture
```

Out[69]=
```
Interval[{-0.00652786, -0.00652786}]
```

Out[70]=
```
Interval[{-0.103692, -0.103692}]
```

Out[71]=
```
Interval[{-0.00642293, -0.00642293}]
```

If the hypothesis of Theorem 3.7 is satisfied, and all of the radii polynomials are negative, then we have a proof of Proposition 3.10(a).

In[72]:=
```
(ρW > ρ[ϵ, ra, rw, rc]) && (Pa[ϵ, ra, rw, rc, ρW] < 0) &&
  (Pw[ϵ, ra, rw, rc, ρW] < 0) && (Pc[ϵ, ra, rw, rc, ρW] < 0) /. WrightConjecture
```
Out[72]=
```
True
```

Proposition 3.10 (b)

We fix a collection of ϵ, r, and ρ for Proposition 3.10 (b).

In[73]:=
```
JonesConjecture = {
   ϵ → Interval[{0.09, 0.09}],
   ra → Interval[{0.1753, 0.1753}],
   rw → Interval[{0.0941, 0.0941}],
   rc → Interval[{0.3829, 0.3829}],
   ρJ → Interval[{1.5940, 1.5940}]};
```

We check that the hypothesis for Proposition B.4 is satisfied

In[74]:=
```
ρ[ϵ, ra, rw, rc] /. JonesConjecture
```
Out[74]=
```
Interval[{1.59369, 1.59369}]
```

In[75]:=
```
ρJ > ρ[ϵ, ra, rw, rc] /. JonesConjecture
```
Out[75]=
```
True
```

We calculate all of the radii polynomials.

In[76]:=
```
Pa[ϵ, ra, rw, rc, ρJ] /. JonesConjecture
Pw[ϵ, ra, rw, rc, ρJ] /. JonesConjecture
Pc[ϵ, ra, rw, rc, ρJ] /. JonesConjecture
```
Out[76]=
```
Interval[{-0.0000298859, -0.0000298859}]
```
Out[77]=
```
Interval[{-0.0000135442, -0.0000135442}]
```
Out[78]=
```
Interval[{-0.0544437, -0.0544437}]
```

If the hypothesis of Theorem 3.7 is satisfied, and all of the radii polynomials are negative, then we have a proof of Proposition 3.10(b).

In[79]:=
```
(ρJ > ρ[ϵ, ra, rw, rc]) && (Pa[ϵ, ra, rw, rc, ρJ] < 0) &&
  (Pw[ϵ, ra, rw, rc, ρJ] < 0) && (Pc[ϵ, ra, rw, rc, ρJ] < 0) /. JonesConjecture
```
Out[79]=
```
True
```

Proposition 3.15

We use numerical optimization (without interval arithmetic) to determine what radii to use.
We fix ϵ=0.10 and try to make r as small as possible.

In[80]:=
```
ϵ316 = 0.10;
numopt316 = NMinimize[{ra + rw + rc,
    {Pa[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2, ρ[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2]] < -10^-10,
     Pw[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2, ρ[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2]] < -10^-10,
     Pc[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2, ρ[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2]] < -10^-10, 0 < ra, 0 < rw, 0 < rc}
   }, {ra , rw, rc }] /. ϵ → .10
```

NMinimize: The following constraints are not valid:

$$\left\{0 < \text{ra}, 0 < \text{rc}, \ll 2\gg, \ll 1\gg, \text{ra}\,\epsilon^2 \left(\frac{16\,\epsilon^2}{5\sqrt{5}\,\pi} + \frac{8\sqrt{1+\ll 1\gg}\;\epsilon\left(\frac{1}{5}\text{Power}[\ll 2\gg] + \text{rw Power}[\ll 2\gg] + \frac{1}{2}\epsilon\,\text{Plus}[\ll 2\gg]\,\text{Plus}[\ll 3\gg]\right)}{5\pi} + \frac{4 \ll 2\gg \left(\frac{1}{5}\text{Power}[\ll 2\gg] + \text{rw} \ll 1\gg + \frac{1}{2} \ll 3\gg\right)}{5\pi^2} + \frac{2\,(1 + 2\,\text{Power}[\ll 2\gg]\,\epsilon)\,(\text{rc Power}[\ll 2\gg] + \epsilon\,\text{Plus}[\ll 2\gg]\,\text{Plus}[\ll 3\gg] + 2\,\text{Plus}[\ll 2\gg]\,\text{Plus}[\ll 2\gg])}{\sqrt{5}\,\pi}\right) + \text{rc}\,\epsilon^2(-1 + \ll 4\gg) + 2\,(\ll 1\gg) + \ll 1\gg < -\frac{1}{\ll 10\gg 0}\right\}$$

. Constraints should be equalities, inequalities, or domain specifications involving the variables.

Out[81]= {0.578203, {ra → 0.0593527, rw → 0.0259709, rc → 0.49288}}

Mathematica's constrained numerical optimization function does not always produce a point for which all the radii polynomials are negative. So we check to see if in fact they are all negative.

In[82]:=
```
Pa[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2, ρ[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2]] /. numopt316[[2]] /. ϵ → ϵ316
Pw[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2, ρ[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2]] /. numopt316[[2]] /. ϵ → ϵ316
Pc[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2, ρ[ϵ, ra ϵ^2, rw ϵ^2, rc ϵ^2]] /. numopt316[[2]] /. ϵ → ϵ316
```

Out[82]= -8.54868×10^{-11}

Out[83]= -8.10034×10^{-11}

Out[84]= -1.38325×10^{-10}

We fix a collection of ϵ, r, and ρ for Proposition 3.15.
Because of interval arithmetic, the r values need to be rounded up slightly.

In[85]:=
```
TightEstimate = {ϵ → Interval[{.10, .10}],
   ra → Interval[{0.0594, 0.0594}],
   rw → Interval[{0.0260, 0.0260}],
   rc → Interval[{0.4929, 0.4929}],
   ρT → Interval[{0.3191, 0.3191}]};
```

We check that the hypothesis for Proposition B.4 is satisfied

In[86]:=
```
ρT > ρ[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc] /. TightEstimate
```

Out[86]= True

We calculate all of the radii polynomials.

In[87]:=
```
Pa[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc, ρT] /. TightEstimate
Pw[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc, ρT] /. TightEstimate
Pc[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc, ρT] /. TightEstimate
```

Out[87]= Interval$\left[\left\{-4.34622\times10^{-7}, -4.34622\times10^{-7}\right\}\right]$

Out[88]= Interval$\left[\left\{-2.70893\times10^{-7}, -2.70893\times10^{-7}\right\}\right]$

Out[89]= Interval$\left[\left\{-1.65154\times10^{-7}, -1.65154\times10^{-7}\right\}\right]$

If the hypothesis of Corollary 3.13 is satisfied, and all of the radii polynomials are negative, then we have a proof of the first half of Proposition 3.15.

In[90]:=
```
(ρT > ρ[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc]) && (Pa[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc, ρT] < 0) &&
  (Pw[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc, ρT] < 0) && (Pc[ϵ, ϵ^2 ra, ϵ^2 rw, ϵ^2 rc, ρT] < 0) /. TightEstimate
```

Out[90]= True

We also make sure that $\alpha>\pi/2$

In[91]:= $\frac{1}{5}\left(\frac{3\pi}{2}-1\right)$ > ra /. TightEstimate

Out[91]= True

Remark 3.15

We check that the r and ρ values in Proposition 3.15 is less than those used in Proposition 3.10 (a) and (b).

In[92]:=
```
(ϵ^2 ra /. TightEstimate) < (ra /. WrightConjecture)
(ϵ^2 rw /. TightEstimate) < (rw /. WrightConjecture)
(ϵ^2 rc /. TightEstimate) < (rc /. WrightConjecture)
(ρT /. TightEstimate) < (ρW /. WrightConjecture)
```

Out[92]= True

Out[93]= True

Out[94]= True

Out[95]= True

In[96]:=
```
(ϵ² ra /. TightEstimate) < (ra /. JonesConjecture)
(ϵ² rw /. TightEstimate) < (rw /. JonesConjecture)
(ϵ² rc /. TightEstimate) < (rc /. JonesConjecture)
(ρT /. TightEstimate) < (ρJ /. JonesConjecture)
```

Out[96]= True

Out[97]= True

Out[98]= True

Out[99]= True

Appendix E: A Priori Estimates on Periodic Orbits

Proposition E.2

We define the function $h_k(\omega)$ as `h[k,w]`.

In[100]:=
```
h[k_, w_] := (k² - 1)/2 w + 2 Sin[w] - 2 k Sin[ k w]
```

If $k = 2$, then clearly $h_k(\omega) > 0$ for $\omega > 4$.
We use interval arithmetic to show that $h_2(\omega)$ is positive for $\omega \in [1.1, 4.0]$.

In[101]:=
```
h[2, Interval[{1.1, 2.0}]] > 0
h[2, Interval[{2.0, 3.0}]] > 0
h[2, Interval[{3.0, 3.5}]] > 0
h[2, Interval[{3.5, 3.7}]] > 0
h[2, Interval[{3.7, 4.0}]] > 0
```

Out[101]= True

Out[102]= True

Out[103]= True

Out[104]= True

Out[105]= True

If $k = 3$, then clearly $h_k(\omega) > 0$ for $\omega > 2$.
We use interval arithmetic to show that $h_2(\omega)$ is positive for $\omega \in [1.1, 2.0]$.

In[106]:=
```
h[4, Interval[{1.1, 2}]] > 0
```

Out[106]= True

If $k = 4$, then clearly $h_k(\omega) > 0$ for $\omega > 4/3$.
We use interval arithmetic to show that $h_2(\omega)$ is positive for $\omega \in [1.1, 1.34]$.

In[107]:=
```
h[4, Interval[{1.1, 1.34}]] > 0
```
Out[107]=
```
True
```

Lemma E.3

In[108]:=
$$\gamma[\epsilon\_, \omega\_] := \frac{1}{2} + \epsilon \left(\frac{2}{3} + \text{Max}\left[\left\{ \frac{\sqrt{2 - 2\,\text{Sin}\left[\omega - \frac{\pi}{2}\right]}}{2}, \frac{2}{3} \right\}\right] \right)$$

Lemma E.4

We define the upper bound b_* to be BB.

In[109]:=
$$\text{BB}[\epsilon\_, \text{ra}\_, \text{rw}\_] := \frac{\pi/2 - \text{rw}}{\pi/2 + \text{ra}} - \frac{1}{2} - \epsilon \left(\frac{2}{3} + \frac{1}{2}\sqrt{2 + 2\,\text{rw}} \right)$$

We compute the values lower and upper bounds -- unscaled

In[110]:=
$$\text{lowB}[\epsilon\_, \text{ra}\_, \text{rw}\_] := \text{BB}[\epsilon, \text{ra}, \text{rw}] - \sqrt{\text{BB}[\epsilon, \text{ra}, \text{rw}]^2 - 2\,\epsilon^2}$$
$$\text{highB}[\epsilon\_, \text{ra}\_, \text{rw}\_] := \text{BB}[\epsilon, \text{ra}, \text{rw}] + \sqrt{\text{BB}[\epsilon, \text{ra}, \text{rw}]^2 - 2\,\epsilon^2}$$

Lemma E.5

We compute a lower bound scaled with ϵ

In[112]:=
$$\text{lowBB}[\epsilon\_, \text{ra}\_, \text{rw}\_] := \frac{1}{\epsilon} \left(\text{BB}[\epsilon, \text{ra}, \text{rw}] - \sqrt{\text{BB}[\epsilon, \text{ra}, \text{rw}]^2 - 2\,\epsilon^2} \right)$$

Section 4: Global Results

Section 4.1: A proof of Wright's Conjecture

Proposition 4.2

The maximum absolute value of a SOPS having $\alpha \in [1.5706, \pi/2]$ is given by μ.

In[113]:=
```
μ = Exp[Interval[{0.04, 0.04}]] - 1
```
Out[113]=
```
Interval[{0.0408108, 0.0408108}]
```

Lemma 4.3

This lemma is concerned with $\alpha \in [1.5706, \pi/2]$.

In[114]:=
$$\alpha\text{Range} = \text{Interval}\left[\left\{1.5706, \frac{\pi}{2}\right\}\right];$$

We show that $0 < (\kappa + 1)\, e^{-\alpha\,\kappa} - 1$ for $\kappa \in [-\mu, 0)$, by showing that the RHS's derivative is negative for $\kappa \in [-\mu, 0)$, and the observation that the RHS equals 0 when κ=0.

In[115]:=
```
D[(κ + 1) E^(-α κ) - 1, κ] /. {α → αRange, κ → -Interval[{0, Max[μ]}]}
```

Out[115]=
```
Interval[{-0.674791, -0.440298}]
```

We define maximum and minimum range of t_+ by `TpMax` and `TpMin` respectively.

In[116]:=
```
TpMax = 2 + 1/αRange;
TpMin = 1 + 1/αRange Log[1 + μ]/μ;
```

We define maximum and minimum range of t_- by `TnMax` and `TnMin` respectively.

In[118]:=
```
TnMax = 3;
TnMin = 1 + 1/αRange;
```

The maximum and minimum period length is given by respectively combining the maximum and minimum values of t_+ and t_-.

In[120]:=
```
Lmin = TpMin + TnMin;
Lmax = TpMax + TnMax;
```

We obtain the appropriate range of frequencies ω by dividing 2π by the period length.

In[122]:=
```
ωMax = 2 π/Lmin
ωMin = 2 π/Lmax
```

Out[122]=
```
Interval[{1.92691, 1.92701}]
```

Out[123]= $\text{Interval}\left[\left\{1.11469, \frac{2\pi}{5 + \frac{2}{\pi}}\right\}\right]$

Lemma 4.5 - First Proof of -- A priori bounds for Wright's Conjecture

This lemma derives bounds on the quantities ω, ϵ_*, $||c||$, and $||K^{-1}c||$ in order.

We show that $\omega \in [1.4219, 1.6887]$.

Lemma 4.3 showed that $\omega \in [1.11, 1.93]$. We will obtain the desired bounds by showing that Line (4.3) of the paper cannot hold for ω in the interval [1.1000, 1.4219] nor in the interval [1.6887, 2.0000].

We define functions equal to the LHS and RHS of Line (4.3) in the paper, and then graph the functions over the relevant ranges.

In[124]:=
```
LHS4p3[ω_, α_] := ω Sqrt[(ω - α)^2 + 2 α ω (1 - Sin[ω])]
RHS4p3[α_] := 2 π/Sqrt[3] α^2 μ (1 + μ)
```

In[126]:=
```
Plot[{Max[LHS4p3[ω, αRange]], Min[RHS4p3[αRange]]}, {ω, 1.1, 2}]
```

Plot: {((ω + Interval[{Times[≪2≫], −1.5706}])^2 + ω Interval[{3.1412, π}] (1 − Sin[ω])) − 0, Im[(ω + Interval[{≪2≫}])^2 + ω Interval[{3.1412, π}] (1 − Sin[≪1≫])] − 0} must be a list of equalities or real−valued functions.

Out[126]=


Essentially, we need to prove that the blue line (LHS) is above the orange line (RHS) for ω in the interval [1.1000, 1.4219] as well the interval [1.6887, 2.0000].

To show Line (4.3) is not satisfied for ω in the interval [1.6887, 2.0000], it suffices to prove the following two points:

- The LHS is greater than the RHS at ω=1.6887 and $\alpha \in [1.5706, \pi/2]$.
- The derivative of the LHS with respect to ω is positive for $\omega \in [1.6887, 2.0000]$ and $\alpha \in [1.5706, \pi/2]$.

We check this below.

In[127]:=
```
LHS4p3[ω, αRange] > RHS4p3[αRange] /. ω → Interval[{1.6887, 1.6887}]
```

Out[127]= True

In[128]:=
```
0 < D[LHS4p3[ω, αRange], ω] /. ω -> Interval[{1.6887, 2.0000}]
```

Out[128]= True

To show Line (4.3) is not satisfied for ω in the interval [1.1000, 1.4219], it suffices to prove the following two points:

- The LHS is greater than the RHS at ω=1.4219 and $\alpha \in [1.5706, \pi/2]$.
- The derivative of the LHS with respect to ω is negative for $\omega \in [1.1000, 1.4219]$ and $\alpha \in [1.5706, \pi/2]$.

We check this below. For the second point, we break the interval [1.1,1.4219] into two intervals [1.10,1.25] and [1.2500,1.4219].

In[129]:=
```
LHS4p3[ω, αRange] > RHS4p3[αRange] /. ω → Interval[{1.4219, 1.4219}]
```

Out[129]= True

In[130]:=
```
0 > D[LHS4p3[ω, αRange], ω] /. ω -> Interval[{1.10, 1.25}]
0 > D[LHS4p3[ω, αRange], ω] /. ω -> Interval[{1.25, 1.4219}]
```
Out[130]= True

Out[131]= True

In conclusion, we have shown that ω must lie in the following interval:

In[132]:=
```
ωRange = Interval[{1.4219, 1.6887}];
```

In[133]:= $\text{Abs}\left[\omega\text{Range} - \frac{\pi}{2}\right]$

Out[133]= Interval[{0, 0.148896}]

We show that $\epsilon_* \le \mu / \sqrt{2}$.

We check that the inequality in Line (4.4) from the paper is true.

In[134]:= $\frac{\pi}{\omega\text{Range}\sqrt{3}}\,\alpha\text{Range}\,\mu\,(1+\mu) \le \frac{2\,\omega\text{Range} - \alpha\text{Range}}{\alpha\text{Range}}$

Out[134]= True

In[135]:=
```
αRange < 2 ωRange
```
Out[135]= True

We conclude with the estimate on ϵ from the paper.

In[136]:= $\epsilon 0 = \mu / \sqrt{2};$

We compute bounds on ||c|| and $\| K^{-1} c \|$.

These are the relevant values for r_α and r_ω.

In[137]:=
```
Abs[π / 2 - αRange] ≤ Interval[{0.02, 0.02}]
Abs[π / 2 - ωRange] ≤ Interval[{0.1489, 0.1489}]
```
Out[137]= True

Out[138]= True

We calculate an upper bound on b_*.

In[139]:= $\text{BB}[\epsilon 0, \text{ra}, \text{rw}] \;/.\; \left\{\text{ra} \to \text{Max}\left[\text{Abs}\left[\alpha\text{Range} - \frac{\pi}{2}\right]\right], \text{rw} \to \text{Max}\left[\text{Abs}\left[\omega\text{Range} - \frac{\pi}{2}\right]\right]\right\}$

Out[139]= Interval[{0.363986, 0.363986}]

We calculate bounds on z_*^+ as in Line (2.10) from the paper (also see Proposition E.4), showing that $z_*^+ \ge 0.72$.

In[140]:=
```
highB[ϵ0, ra, rw] /. {ra → Max[Abs[αRange - π/2]], rw → Max[Abs[ωRange - π/2]]}
```

Out[140]=
```
Interval[{0.725677, 0.725677}]
```

We compute an upper bound on $\|\tilde{c}\|$ using Line (4.2), which is less than 0.09.

In[141]:=
```
π/(ωRange Sqrt[3]) αRange μ (1 + μ) < Interval[{0.09, 0.09}]
```

Out[141]=
```
True
```

We calculate bounds on z_*^-/ϵ as in Lemma E.5, showing that $\|c\| \leq 0.0796$.

In[142]:=
```
lowBB[ϵ0, ra, rw] /. {ra → Max[Abs[αRange - π/2]], rw → Max[Abs[ωRange - π/2]]}
```

Out[142]=
```
Interval[{0.0795328, 0.0795328}]
```

We bound $\|K^{-1} c\|$ using Lemma 2.5(b).

In[143]:=
```
ρBound = (2 + lowBB[ϵ0, ra, rw]^2) / BB[ϵ0, ra, rw] /.
  {ra → Max[Abs[αRange - π/2]], rw → Max[Abs[ωRange - π/2]]}
```

Out[143]=
```
Interval[{5.51209, 5.51209}]
```

We adjust our bound on $\|K^{-1} c\|$ to be independent of ϵ.

In[144]:=
```
ϵ0 ρBound
```

Out[144]=
```
Interval[{0.159066, 0.159066}]
```

Theorem 4.6 - Proof of Wright's Conjecture

We use numerical optimization (without interval arithmetic) to suggest what radii to use in Proposition 3.10(a).
This (non-rigorous) calculation is not explicitly used in the paper.
This calculation does not appear earlier because some of the constraints are the result calculations made in Section 4.

In[145]:=
```
ϵ311a = 0.029;
ra311a = 0.0002 + Da[ϵ311a, 0];
rw311a = 0.15 + Dw[ϵ311a, 0];
rc311a = 0.08 + δ[ϵ311a, 0];
```

In[149]:=
```
numopt311a = NMaximize[
   {rw,
    Pa[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] < 0,
    Pw[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] < 0,
    Pc[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] < 0,
    ra >  ra311a,
    rw > rw311a,
    rc > rc311a, rc > 0, ρ[ϵ, ra, rw, rc] > 0},
   {ra , rw , rc }] /. ϵ → ϵ311a
```

NMaximize: The following constraints are not valid: {≪1≫}. Constraints should be equalities, inequalities, or domain specifications involving the variables.

NMaximize: NMaximize was unable to generate any initial points satisfying the inequality constraints $\left\{\ll 1\gg \leq 0,\ 0.00030142 + \ll 27\gg + \frac{\ll 1\gg}{\ll 1\gg} + \frac{0.000680095\ \mathrm{ra} \ll 1\gg \mathrm{rw}^2}{1 - 2.19598\,(\ll 1\gg)} \leq 0,\ \ll 1\gg \leq 0,\ 2.81838\times 10^{-6} - 0.9983\ \mathrm{ra} + \ll 27\gg + \frac{0.00131561\ \mathrm{ra}\ \mathrm{rc}^2\ \mathrm{rw}^2}{1 - 2.19598\,(0.0001682 + \mathrm{rw})} \leq 0\right\}$. The initial region specified may not contain any feasible points. Changing the initial region or specifying explicit initial points may provide a better solution.

Out[149]= {0.177076, {ra → 0.132361, rw → 0.177076, rc → 0.175244}}

Mathematica's constrained numerical optimization function does not always produce a point for which all the radii polynomials are negative. So we check to see if in fact they are all negative.

In[150]:=
```
Pa[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] /. numopt311a[[2]] /. ϵ → ϵ311a
Pw[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] /. numopt311a[[2]] /. ϵ → ϵ311a
Pc[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] /. numopt311a[[2]] /. ϵ → ϵ311a
```

Out[150]= -5.05778×10^{-9}

Out[151]= -0.105995

Out[152]= -1.3752×10^{-9}

We check that the set S is contained inside $B_{\epsilon}(r, \rho)$. The terms on the left are from Proposition 3.10(a).

In[153]:=
```
WrightConjecture
```

Out[153]= {ϵ → Interval[{0.029, 0.029}], ra → Interval[{0.13, 0.13}], rw → Interval[{0.17, 0.17}], rc → Interval[{0.17, 0.17}], ρW → Interval[{1.78, 1.78}]}

In[154]:=
```
(*rα*) Interval[{0.13, 0.13}] ≥ Interval[{0.0002, 0.0002}] + Da[ϵ0, 0]
(*rω*) Interval[{0.17, 0.17}] ≥ Interval[{0.15, 0.15}] + Dw[ϵ0, 0]
(*rc*) Interval[{0.17, 0.17}] ≥ Interval[{0.08, 0.08}] + δ[ϵ0, 0]
(*ρ*) Interval[{1.78, 1.78}] ≥ Interval[{0.08, 0.08}]
```

Out[154]= True

Out[155]= True

Out[156]= True

Out[157]= True

Appendix F: Implicit Differentiation

Lemma F.2

We define $\hat{f}_{\epsilon,1}$ as **fϵ1**.

In[158]:=
```
fϵ1[ϵ_, da_, dw_, rc_, δ0_] :=
```
$$\frac{1}{2}\,\delta0\left(\sqrt{2}\;da + 3\,dw\left(\frac{\pi}{2} + da\right)\right) + rc\left(\frac{\pi}{2} + da\right)\left(1 + \delta0 + \frac{1}{2}\,rc\right)$$

We define $\hat{f}_{\epsilon,c}$ as **fϵc**.

In[159]:=
```
fϵc[ϵ_, da_, dw_, δ_, rc_, δ0_] :=
```
$$\frac{2}{\pi\sqrt{5}}\left(2\left(da + \frac{\pi}{2}\,dw\right) + \delta0\left(\sqrt{2}\;da + 3\,dw\left(\frac{\pi}{2} + da\right)\right) + \left(\frac{\pi}{2} + da\right)\left(4\,rc + \delta^2\right)\right)$$

Prior Estimates

We recall our estimates from Proposition 3.16

In[160]:=
```
ϵLocal = ϵ /. TightEstimate;
raLocal = ra /. TightEstimate;
rwLocal = rw /. TightEstimate;
rcLocal = rc /. TightEstimate;
ρLocal = ρT /. TightEstimate;
RA[ϵϵ_] := raLocal ϵϵ^2
RW[ϵϵ_] := rwLocal ϵϵ^2
RC[ϵϵ_] := rcLocal ϵϵ^2
```

We define the upper bound $\mathcal{Z}_\epsilon$ from Equation (4.6) as **zz**.

In[168]:=
```
Zϵ[ϵ_] := Z[ϵ, RA[ϵ], RW[ϵ], RC[ϵ], ρLocal]
```

Corollary F.3

We define our upper bound on $A^\dagger\left(\frac{\partial F}{\partial \epsilon} - \Gamma\right)$ as **AFminusGamma**

In[169]:=
```
AFminusGamma[ϵ_] := (IdentityMatrix[3] + ϵ A0A1).
```
$$\left\{\left(1 + \frac{4}{\pi^2}\right)^{1/2} f\epsilon1[\epsilon,\, Da[\epsilon,\, RA[\epsilon]],\, Dw[\epsilon,\, RW[\epsilon]],\, RC[\epsilon],\, \delta[\epsilon,\, 0]],\right.$$
$$\frac{2}{\pi}\, f\epsilon1[\epsilon,\, Da[\epsilon,\, RA[\epsilon]],\, Dw[\epsilon,\, RW[\epsilon]],\, RC[\epsilon],\, \delta[\epsilon,\, 0]],$$
$$\left. f\epsilon c[\epsilon,\, Da[\epsilon,\, RA[\epsilon]],\, Dw[\epsilon,\, RW[\epsilon]],\, \delta[\epsilon,\, RC[\epsilon]],\, RC[\epsilon],\, \delta[\epsilon,\, 0]]\right\}$$

Lemma F.4

We define $|\alpha'|$ as **Aprime**, $|\omega'|$ as **Wprime**, and (a bound on) $\|c'\| = 2\,|c'_2| + 2\,|c'_3|$ as **Cprime**.

In[170]:=
```
Aprime[ϵ_] := 2/5 (3 π/2 - 1) ϵ
Wprime[ϵ_] := 2/5 ϵ
Cprime[ϵ_] := 2/Sqrt[5] + 2/Sqrt[10] ϵ + 18/(5 Sqrt[50]) ϵ^2
```

We define Q_ϵ^0 as **Qϵ0**

In[173]:=
```
Qϵ0[ϵ_] := {Aprime[ϵ], Wprime[ϵ], Cprime[ϵ]}
```

We define Q_ϵ as **Qϵ**

In[174]:=
```
Qϵ[ϵ_] := Qϵ0[ϵ] + AFminusGamma[ϵ]
```

Lemma F.5

We define M_ϵ and M'_ϵ as **Mϵ** and **MMϵ**.

In[175]:=
```
Mϵ[ϵ_] := (1/ϵ^2 AFminusGamma[ϵ])[[1]]
MMϵ[ϵ_] := (1/ϵ^2 Zϵ[ϵ].Inverse[IdentityMatrix[3] - Zϵ[ϵ]].Qϵ[ϵ])[[1]]
```

Section 4: Global Results

Section 4.2: Work towards proving Jones' Conjecture

Theorem 4.7

We compute if ϵ=0.1, then how big α must be, accounting for error given in TightEstimate.

In[177]:=
```
a[ϵ] - π/2 /. ϵ → Interval[{0.1, 0.1}]
a[ϵ] - π/2 - RA[ϵ] /. ϵ → Interval[{0.1, 0.1}]
```

Out[177]=
```
Interval[{0.00742478, 0.00742478}]
```

Out[178]=
```
Interval[{0.00683078, 0.00683078}]
```

We check that α'>0 when ϵ =0.1

We check the last inequality stated in the proof to see if it is greater than 0.

In[179]:=
$$0 < \frac{2}{5}\left(\frac{3\pi}{2} - 1\right)\epsilon - \epsilon^2\,(\text{M}\epsilon[\epsilon] + \text{MM}\epsilon[\epsilon])\ /.\ \epsilon \to \text{Interval}[\{0.1, 0.1\}]$$

Out[179]= True

Theorem 4.9 -- Jones Conjecture

Finding some radii to be used in Proposition 3.10(b)

We use numerical optimization (without interval arithmetic) to suggest what radii to use in Proposition 3.10(b).
This (non-rigorous) calculation is not explicitly used in the paper.
This calculation does not appear earlier because some of the constraints are the result calculations made in Section 4.

In[180]:=
```
raOpt[ϵϵ_] := Min[raLocal] (ϵϵ)^2
```

In[181]:=
```
ϵ311b = 0.09;
numopt311b = NMaximize[{rw,
    Pa[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] < -10^-10,
    Pw[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] < -10^-10,
    Pc[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] < -10^-10,
    rc == δ[ϵ, 0] + lowBB[ϵ, Da[ϵ, -raOpt[ϵ]], rw - Dw[ϵ, 0]],
    ra > 0, rw > 0, rc > 0, ρ[ϵ, ra, rw, rc] > 0},
   {ra , rw , rc }] /. ϵ → ϵ311b
```

NMaximize: The following constraints are not valid:

$$\left\{\frac{2\,\epsilon}{\sqrt{5}} + \frac{-\frac{1}{2} + \frac{\frac{\pi}{2} - \text{rw} + \frac{1}{5}\,\text{Power}[\ll 2 \gg]}{\text{Times}[\ll 2 \gg] + \text{Times}[\ll 2 \gg] + \text{Times}[\ll 3 \gg]} - \epsilon\left(\frac{2}{3} + \frac{1}{2}\,\text{Power}[\ll 2 \gg]\right) - \sqrt{\text{Times}[\ll 2 \gg] + \text{Power}[\ll 2 \gg]}}{\epsilon} == \text{rc}, \ll 6 \gg, \ll 1 \gg\right\}.$$

Constraints should be equalities, inequalities, or domain specifications involving the variables.

NMaximize: NMaximize was unable to generate any initial points satisfying the inequality constraints

$$\left\{\ll 1 \gg, 0.000101752 + \ll 29 \gg \le 0, \ll 1 \gg \le 0, 0.000261583 - 0.982698\,\text{ra} + 0.0257553\,\text{rc} + \ll 25 \gg + \frac{0.0222804\,\text{rc}^2\,\text{rw}^2}{1 - 2.52928\,(0.00162 + \text{rw})} + \frac{0.01413\,\text{ra}\,\text{rc}^2\,\text{rw}^2}{1 - 2.52928\,(0.00162 + \text{rw})} \le 0\right\}.$$

The initial region specified may not contain any feasible points. Changing the initial region or specifying explicit initial points may provide a better solution.

Out[182]= {0.0942018, {ra → 0.175485, rw → 0.0942018, rc → 0.382888}}

Mathematica's constrained numerical optimization function does not always produce a point for which all the radii polynomials are negative. So we check to see if in fact they are all negative.

In[183]:=
```
Pa[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] /. numopt311b[[2]] /. ϵ → ϵ311b
Pw[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] /. numopt311b[[2]] /. ϵ → ϵ311b
Pc[ϵ, ra, rw, rc, ρ[ϵ, ra, rw, rc]] /. numopt311b[[2]] /. ϵ → ϵ311b
```

Out[183]= -3.33175×10^{-9}

Out[184]= -1.40111×10^{-9}

Out[185]= -0.0540937

Given the radii in Proposition 3.10 (b), we derive the needed values for Theorem 4.9

Below are computations used in determining the numbers given in the statement of the proposition.

We fix the ϵ we are using.

In[186]:=
```
ϵJ = ϵ /. JonesConjecture
```

Out[186]= Interval[{0.09, 0.09}]

We check to see how large a range of α we could hope to prove using $\epsilon = 0.09$

In[187]:=
$$\alpha\text{JRange} = \text{a}[\epsilon\text{J}] - \frac{\pi}{2} - \text{RA}[\epsilon\text{J}]$$

Out[187]= Interval[{0.00553293, 0.00553293}]

Given the r_w in Proposition 3.10(b), we calculate how far away ω can be from $\pi/2$.

In[188]:=
```
ωJRange = rw - Dw[ϵ, 0] /. JonesConjecture
```

Out[188]= Interval[{0.09248, 0.09248}]

Given the r_α, r_ω above, we calculate how big to expect $\| c \|$ to be.

In[189]:=
```
smallCmax = lowBB[ϵJ, ra, rw] /. {ra → αJRange, rw → ωJRange}
```

Out[189]= Interval[{0.302317, 0.302317}]

We calculate ρ

In[190]:=
```
BB[ϵJ, ra, rw] /. {ra → αJRange, rw → ωJRange}
```

Out[190]= Interval[{0.311305, 0.311305}]

The following is the coefficient on ϵ which bounds $\| K^{-1} c \|$.

In[191]:=
```
ρBound = (2 + smallCmax^2) / BB[ϵJ, ra, rw] /. {ra → αJRange, rw → ωJRange}
```

Out[191]= Interval[{6.71816, 6.71816}]

We adjust our bound on $\| K^{-1} c \|$ to be independent of ϵ.

In[192]:=
```
ϵJ ρBound
```

Out[192]= Interval[{0.604634, 0.604634}]

We show $\tilde{S} \subset B_\epsilon(r, \rho)$.

In[193]:=
```
JonesConjecture
```

Out[193]=
```
{ϵ → Interval[{0.09, 0.09}],
 ra → Interval[{0.1753, 0.1753}], rw → Interval[{0.0941, 0.0941}],
 rc → Interval[{0.3829, 0.3829}], ρJ → Interval[{1.594, 1.594}]}
```

In[194]:=
```
(*rα*) Interval[{0.1753, 0.1753}] ≥ Interval[{0, 0.00553}] + Da[ϵJ, 0]
(*rω*) Interval[{0.0941, 0.0941}] ≥ Interval[{0.0924, 0.0924}] + Dw[ϵJ, 0]
(*rc*) Interval[{0.3829, 0.3829}] ≥ Interval[{0.30232, 0.30232}] + δ[ϵJ, 0]
(*ρ*) Interval[{1.594, 1.594}] ≥ Interval[{0.32, .32}]
```

Out[194]= True

Out[195]= True

Out[196]= True

Out[197]= True

We check the next relevant inequalities.

In[198]:=
```
Dw[ϵJ, 0] + RW[ϵJ] ≤ Interval[{0.0924, 0.0924}]
```

Out[198]= True

In[199]:=
```
δ[ϵJ, 0] + RC[ϵJ] ≤ Interval[{0.30232, 0.30232}]
```

Out[199]= True

Theorem 4.10

We do a calculation indicative of what L_2 value to prove the theorem for.

In[200]:= $(2\,\epsilon J)\sqrt{\frac{6\,\omega}{\pi}}$ `/. ω -> ` $\frac{\pi}{2}$ `+ Interval[{-0.0924, 0.0924}]`

Out[200]=
```
Interval[{0.30246, 0.320808}]
```

We use Lemma 4.1 to check that the Fourier coefficients satisfy the bounds in Theorem 4.10

In[201]:= $\sqrt{\left(\frac{\pi}{6\,\omega}\right)}$ `Interval[{0.302, 0.302}] ≤ Interval[{0.18, 0.18}] /.` `ω -> ` $\frac{\pi}{2}$ `+ Interval[{-0.0924, 0.0924}]`

Out[201]= True

We check additional inequalities.

In[202]:= `α < 2 ω /. {α → ` $\frac{\pi}{2}$ `+ Interval[{0, 0.00553}], ω -> ` $\frac{\pi}{2}$ `+ Interval[{-0.0924, 0.0924}]}`

Out[202]= True

In[203]:=
```
Interval[{0.174, 0.174}] ≤ (2 ω - α)/α /.
  {α → π/2 + Interval[{0, 0.00553}], ω -> π/2 + Interval[{-0.0924, 0.0924}]}
```

Out[203]= True

We calculate bounds on z_*^+ as in Line (2.10) from the paper (also see Proposition E.4), showing that $z_*^+ \geq 0.595$.

In[204]:=
```
highB[ϵ, ra, rw] /. {ϵ → Interval[{0.09, 0.09}],
   ra → Interval[{0.00553, 0.00553}], rw → Interval[{0.0924, 0.0924}]}
```

Out[204]= Interval[{0.595516, 0.595516}]

We calculate bounds on z_*^- / ϵ as in Lemma E.5, showing that $\|c\| \leq 0.30226$.

In[205]:=
```
lowBB[ϵ, ra, rw] /. {ϵ → Interval[{0.09, 0.09}],
   ra → Interval[{0.00553, 0.00553}], rw → Interval[{0.0924, 0.0924}]}
```

Out[205]= Interval[{0.302259, 0.302259}]